Q:

Help me export high-resolution TIFF figure

I have generated a figure in MMA by ContourPlot3D. How can I export this figure in high-resolution TIFF format(say, 1200 dpi for lines, 300 dpi for halftone and color in CMYK format, and 600 dpi for a mixture of lines and halftone).
sf1[x_, y_, z_] := x *(1/(1 + 0.06))^3 + y*(1/(1 + 0.06))^2 - z/3;
sf2[x_, y_, z_] := x *(1 + 5*0.06)*(1/((1 + 5 0.06) + 0.06))^3 + 
y*(1 + 5 0.06)*(1/((1 + 5 0.06) + 0.06))^2 - z/3; 
sf3[x_, y_, z_] := x*(1 + 10 0.06)*(1/((1 + 10 0.06) + 0.06))^3 + 
y*(1 + 10 0.06)*(1/((1 + 10 0.06) + 0.06))^2 - z/3; 
pla = ContourPlot3D[{sf1[x, y, z] == 0, sf2[x, y, z] == 0, sf3[x, y, z] == 0}, {x, 0, 200}, {y, 0, 120}, {z,
0, 210}, Ticks -> None, Mesh -> None, 
BoundaryStyle -> Directive[Black, Thickness[0.004]], Boxed -> False,
AxesOrigin -> {0, 0, 0},
ImageSize -> 350,
ImagePadding -> 20,
ContourStyle -> {Directive[Green], Directive[Opacity[.6], Red], 
Directive[Opacity[.5], Blue]}
, Lighting -> "Neutral"]

A:

I don't know a way to export a figure with different resolutions for different elements, the term "resolution" normally applies to the whole figure. You have a 350 printer's points wide figure which you seemingly wish to export with resolution 1200 dpi. This means that you wish to export a figure with width 
Round[1200*(350/72)]

5833

pixels. Not every graphical card allows to render such a large figure at once but mine allows:
Export["pla.tiff", pla, ImageResolution -> 1200, "ColorSpace" -> CMYKColor]

Import["pla.tiff", {{"ImageSize", "ColorSpace"}}]

{{5833, 6276}, CYMKColor}

The result is a TIFF file of size 139 Mb (the CYMKColor colorspace reported seems to be a bug: actually it is CMYKColor).
If your graphical card does not allows this or if you do not have sufficient volume of physical memory you need to apply an incremental rendering technique. This post of mine (see the "EDIT 4" section) shows Inset-based approach, and recent post by Vitaliy Kaurov (search for "3dImagePartitionExport.nb") in this Wolfram Community thread demonstrates another technique. 
P.S. For some reason MMa 10.1.0 exports a flat black image while MMa 8.0.4 exports correct image when ImageResolution -> 1200. With version 10.1.0 I get the correct image up to ImageResolution -> 1000.

